
Abstract
This is a tutorial for mathematically inclined functional program-
mers, based on previously published, peered reviewed theoretical
work. We discuss a higher-type functional, written here in the func-
tional programming language Haskell, which (1) optimally plays
sequential games, (2) implements a computational version of the
Tychonoff Theorem from topology, and (3) realizes the Double-
Negation Shift from logic and proof theory. The functional makes
sense for finite and infinite (lazy) lists, and in the binary case it
amounts to an operation that is available in any (strong) monad.
In fact, once we define this monad in Haskell, it turns out that this
amazingly versatile functional is already available in Haskell, in the
standard prelude, called sequence, which iterates this binary op-
eration. Therefore Haskell proves that this functional is even more
versatile than anticipated, as the function sequencewas introduced
for other purposes by the language designers, in particular the itera-
tion of a list of monadic effects (but effects are not what we discuss
here).

Categories and Subject Descriptors D.1.1 [Programming tech-
niques]: functional programming

General Terms Algorithms, languages, theory.

Keywords Functional programming, Haskell, monad, search,
game theory, optimal strategy, exhaustible set, axiom of choice,
infinite data, dependent type, Agda, topology, logic, foundations.

1. An amazingly versatile functional
Perhaps the most concise and self-contained definition of the func-
tional we discuss in this tutorial is this, using Haskell notation:

bigotimes :: [(x -> r) -> x] -> ([x] -> r) -> [x]
bigotimes [] p = []
bigotimes (e : es) p = x0 : bigotimes es (p.(x0:))

where x0 = e(\x -> p(x : bigotimes es (p.(x:))))

This is an example of an algorithm that is patently not self-
explanatory. Here is a preliminary indication of what is going on,
to be elaborated below. Think of r as a type of generalized truth

values. A particular case of interest will be r = Bool, but not the
only relevant one. The input of the algorithm is a list of so-called
selection functions (x -> r) -> x for a type x, and the output
is a single, combined selection function ([x] -> r) -> [x] for
the type [x] of lists of elements of the type x. If the input list of
selection functions is

ε = [ε0, ε1, . . . , εn, . . . ],

then the mathematical notation for the output of the algorithm isO
i

εi.

As we shall see, the definition of the algorithm amounts to, and can
in fact be literally written as a recursive definition,O

i

εi = ε0 ⊗
O

i

εi+1,

for a suitable binary operation⊗ that combines two selection func-
tions.

This algorithm makes sense for both finite and infinite (lazy)
lists, and we shall write it in several different but equivalent ways.
It is difficult to give a general specification of what it does, because
it turns out that it performs a variety of seemingly unrelated tasks:

1. It optimally plays sequential games.

2. It implements a computational manifestation of the Tychonoff
Theorem from topology.

3. It realizes the Double-Negation Shift (DNS) from logic and
proof theory (whose role is to computationally extract witnesses
from classical proofs that use the axiom of countable choice).

And more, as discussed in [6–8, 10–12]. For example, the so-called
Bekic’s Lemma for fixed-points that arises in domain theory and
programming language semantics is a particular case of the product
of selection functions [8].

Originally, the algorithm was designed to achieve (2), but, by
studying it in more detail, we discovered that it also achieves (3)
and (1), in chronological order. We regard this as a rather surprising
course of events, given that the Tychonoff Theorem hasn’t appeared
so far in the theoretical study of (1) and (2), at least not explicitly.
In any case, it is certainly surprising that the same mathematical op-
eration or algorithm should simultaneously solve these seemingly
unrelated tasks, each of which is fundamental in its own right.

It is the purpose of this tutorial to explain the main ideas behind
the mathematical operation

N
, defined and studied in [6–8, 10–

12]. This expository paper is an invitation for the reader to explore
these references, which give full mathematical details and proofs,
as well as more techniques, ideas and directions.



We also implement a number of sample applications in func-
tional programming, such as playing Tic-Tac-Toe (aka Noughts-
and-Crosses), solving the n-Queens puzzle, and deciding equality
of functions (!) on certain infinite types, among others.

Organization. 2. Selection functions. 3. Products of selection
functions. 4. Playing games. 5. The Tychonoff Theorem. 6. Mon-
ads. 7. The Double-Negation Shift. 8. Concluding remarks.

2. Selection functions
If we define

type J r x = (x -> r) -> x

then the type specification of the function bigotimes can be
rewritten as

bigotimes :: [J r x] -> J r [x]

To understand the type constructor J, we also consider

type K r x = (x -> r) -> r

The type constructor K r is known as the continuation monad,
and is related to control operators such as call/cc. But we shall
instead regard it as a type of generalized quantifiers over the type x,
with type of generalized truth values r. As we shall see later, J r
is a monad too, called the selection monad, and there is a monad
morphism to K r. We shall have occasion to use the morphism
before we give J r the structure of a monad:

overline :: J r x -> K r x
overline e = \p -> p(e p)

The Haskell notation
overline e

corresponds to the mathematical notation

ε.

We refer to the elements of J r x as selection functions. What
overline does, then, is to transform selection functions into quan-
tifiers. In this context, we refer to the elements of the function type
(x -> r) as predicates.

Terminology Mathematics Haskell Type
predicate p, q p, q x -> r

selection function ε, δ e, d J r x
quantifier φ, γ phi, gamma K r x

With r = Bool, a particular element of the type K r x is the ex-
istential quantifier, or the existential quantification functional to
be more precise, which arises when we consider selection func-
tions for sets. After considering this particular case of a selection
function, we consider the more general case of selection functions
for arbitrary quantifiers, which include the universal quantifier, of
course, but other generalized quantifiers as well.

In this discussion we adopt the traditional terminology from
logic that refers to higher-type functions as functionals, and we
stress that in logic one usually adopts the terminology higher-
order to logics that quantify over propositions, and the terminology
higher-type to formal systems that allow nested function types
(such as Gödel’s system T and generalizations, including PCF and
Haskell).

2.1 Selection functions for sets
A selection function for a set S finds an element of S for which
a given predicate holds. But we emphasize that we require our
selection functions to be total. Hence if there is no element of S
satisfying the predicate, we select an arbitrary element of S, and so

S must be non-empty. Here is an example, where we consider finite
sets given as finite lists:

find :: [x] -> J Bool x
find [] p = undefined
find [x] p = x
find (x:xs) p = if p x then x else find xs p

forsome, forevery :: [x] -> K Bool x
forsome = overline.find
forevery xs p = not(forsome xs (not.p))

Or, expanding the definitions,

forsome xs p = p(find xs p)

Our definition of the existential quantifier is the same as in Hilbert’s
ε-calculus:

∃x p(x) ⇐⇒ p(ε(p)).

Notice that the definition of forevery uses the De Morgan Law
for quantifiers,

∀x p(x) ⇐⇒ ¬∃x ¬p(x).
As discussed above, we are interested in finite non-empty lists only,
to make sure the produced selection function is total. For instance,

find [1..100] (\x -> odd x && x > 17) = 19
forsome [1..100] (\x -> odd x && x > 17) = True

find [1..100] (\x -> odd x && even x) = 100
forsome [1..100] (\x -> odd x && even x) = False

A selection function ε for a set S has to satisfy:

1. ε(p) ∈ S, whether or not there actually is some x ∈ S such
that p(x) holds.

2. If p(x) holds for some x ∈ S, then it holds for x = ε(p).

Notice that the first condition forces the set S to be non-empty.
Here is another example that we shall need later:

findBool :: J Bool Bool
findBool p = p True

This is equivalent to

findBool p = if p True then True else False

but it is silly to check the two possible cases. This is so both
conceptually and for the sake of efficiency.

2.2 Selection functions for quantifiers
Our use of selection functions goes beyond the above idea. If φ(p)
stands for ∃x ∈ S p(x), then Hilbert’s condition can be written as
an equation,

φ(p) = p(ε(p)),

or equivalently, using the monad morphism, as the equation

φ = ε.

If this equation holds, we say that ε is a selection function for the
quantifier φ. Thus, a selection function for a set S is the same thing
as a selection function for the existential quantifier of S.

When φ(p) is the universal quantifier ∀x ∈ S p(x) of the set S,
the above equivalent equations amount to

1. ε(p) ∈ S.

2. If p(x) holds for x = ε(p), then it holds for all x ∈ S.

This is known as the Drinker Paradox: in every pub there is a
person a such that if a drinks then everybody drinks. Here S is



the set of people in the pub, and p(x) means that x drinks, and we
calculate a with the selection function as a = ε(p). A selection
function for the universal quantifier of a finite set can be defined in
Haskell as follows:

findnot :: [x] -> J Bool x
findnot [] p = undefined
findnot [x] p = x
findnot (x:xs) p = if p x then findnot xs p else x

Notice that this satisfies

findnot xs p = find xs (not.p)

and that the function forevery defined above satisfies

forevery = overline.findnot

We now consider “predicates” that give numbers rather than
boolean truth values (sometimes known as objective functions). For
example, the predicate may assign prices to goods. Given such a
predicate, we may wish to

1. find the price of the most expensive good — this is done by a
quantifier, called sup,

2. find the most expensive good — this is done by a selection
function, called argsup.

Of course, the result of the selection function is ambiguously de-
fined, because there may be more than one good with the high-
est price (just as there may be more than one element satisfying a
given boolean-valued predicate, and we chose one arbitrarily with
the above algorithms).

This should be compared with the Maximum-Value Theorem,
which says that any continuous function p : [0, 1] → R attains its
maximum value. This means that there is a ∈ [0, 1] such that

sup p = p(a).

However, the proof is non-constructive, and it is known that there
is no computable selection function argsup that calculates a =
argsup(p) (but the supremum quantifier sup: ([0, 1] → R) → R
is known to be computable). This can also be compared to the
Mean-Value Theorem, which says that for some a ∈ [0, 1],Z

p = p(a).

Again this a cannot be found from p using an algorithm. Similarly
the Drinker Paradox can be written

∀(p) = p(a),

and finding elements in sets corresponds to

∃(p) = p(a).

We have already seen that we can find a as a = ε(p) if the
ranges of the universal and existential quantifiers are finite (and
we’ll go beyond the finite case below). The general equation we
are considering here is thus

φ(p) = p(a),

where in favourable circumstances a can be calculated as

a = ε(p)

for a suitable selection function ε that works for all predicates p.
Continuing the discussion about prices of goods, in the next

example we artificially assume that the only two goods are True
and False, and that our numbers are integers:

argsup :: J Int Bool
argsup p = p True > p False

sup :: K Int Bool
sup = overline argsup

Of course the definition of argsup is equivalent to

argsup p = if p True > p False then True else False

which is perhaps more natural. If we want our goods to be given as
a list, we can instead define

argsup :: [x] -> J Int x
argsup [] p = undefined
argsup [x] p = x
argsup (x:y:zs) p = if p x < p y

then argsup (y:zs) p
else argsup (x:zs) p

This can be made more efficient by avoiding re-evaluations of p.
Moreover, if the range of p is not the set of all integers, but just a
finite range, such as −1, 0, 1, then the following is a more efficient
algorithm, which stops when the maximum value 1 is reached, and
switches to a specialized algorithm when the value 0 is reached:

argsup :: [x] -> J Int x
argsup [] p = undefined
argsup (x:xs) p = f xs x (p x)
where f xs a 1 = a

f [] a r = a
f (x:xs) a (-1) = f xs x (p x)
f xs a 0 = g xs
where g [] = a

g (x:xs) | p x == 1 = x
| otherwise = g xs

Infima can he handled by mirroring this algorithm, or alternatively
by reduction to suprema, where the involution \x -> -x plays the
role of negation in this context:

arginf :: [x] -> J Int x
arginf xs p = argsup xs (\x -> - p x)

These two functions will be useful for two-player games in which
there can be a draw.

3. Products of selection functions
We first consider binary products of selection functions, and then
iterate them (in)finitely often, obtaining the functional discussed in
Section 1.

3.1 Binary product
In order to motivate our construction, consider the following vari-
ation of the Drinker Paradox discussed above: in every pub there
are a man a0 and a woman a1 such that if a0 buys a drink to a1

then every man buys a drink to some woman. If the sets of men and
women are X0 and X1 respectively, and if we define the combined
quantifier φ = ∀ ⊗ ∃ by

φ(p) = (∀x0 ∈ X0 ∃x1 ∈ X1 p(x0, x1)) ,

then this can be formalized by saying that, for a suitable pair
a = (a0, a1) ∈ X0 ×X1,

φ(p) = p(a).

Our objective is to calculate such a pair a.
The key observation is that we have selection functions for the

quantifiers ∀ and ∃, say ε0 and ε1, and that what we need is a
suitable combined selection function ε = ε0⊗ε1 for the combined
quantifier φ = ∀ ⊗ ∃. We define combinations of quantifiers and
of selection functions in such a way that they commute with the



above monad morphism, where the combinator is written ⊗ for
both quantifiers and selection functions. In other words:

If εi is a selection function for the quantifier φi for i = 0, 1,
then ε = ε0 ⊗ ε1 is a selection function for the quantifier
φ = φ0 ⊗ φ1.

It is easy to define the product ⊗ of quantifiers, generalizing from
the above example, where φ = φ0 ⊗ φ1 for φ0 = ∀ and φ1 = ∃:

(φ0 ⊗ φ1)(p) = φ0(λx0.φ1(λx1.p(x0, x1))).

The definition of the product of selection functions is a bit subtler:

(ε0 ⊗ ε1)(p) = (a0, a1)

where a0 = ε0(λx0.ε1(λx1.p(x0.x1)))

a1 = ε1(λx1.p(a0, x1)).

What we need to check is that

ε0 ⊗ ε1 = ε0 ⊗ ε1,

which amounts to

if φi = εi for i = 0, 1, then φ0 ⊗ φ1 = ε0 ⊗ ε1.

This is a routine verification that the reader is encouraged to per-
form for the sake of understanding (and that is performed in the
given references). Coming back to our motivating example, the re-
quired man and woman can be calculated with the formula

(a0, a1) = (ε0 ⊗ ε1)(p),

where ε0 and ε1 are selection functions for the quantifiers ∀ and ∃
respectively.

The binary product of selection functions can be implemented
as

otimes :: J r x0 -> J r x1 -> J r (x0,x1)
otimes e0 e1 p = (a0,a1)
where a0 = e0(\x0 -> overline e1(\x1 -> p(x0,x1)))

a1 = e1(\x1 -> p(a0,x1))

We leave the implementation of the binary product of quantifiers
as an exercise (but later we shall implement the binary product
for all monads, which will apply to the selection monad and the
continuation monad in particular).

3.2 Iterated product
Given a sequence of sets X0, X1, . . . , Xn, . . . , define the productQ

i<n Xi by induction on n asY
i<0

Xi = {()},
Y

i<n+1

Xi = X0 ×
Y
i<n

Xi+1.

Informally, Y
i<n

Xi = X0 × · · · ×Xn−1.

For each n we define a functionO
:
Y
i<n

JRXi → JR
Y
i<n

Xi,

by induction asO
i<0

εi = λp.(),
O

i<n+1

εi = ε0 ⊗
O
i<n

εi+1.

Finite products of quantifiers can be defined in the same way, and
the following equation holds, by induction:O

i<n

εi =
O
i<n

εi.

Of course, the products are of quantifiers in the left-hand side and
of selection functions in the right-hand side.

In dependently typed languages such as Agda, one can write
a single product program that works for every n, but in Haskell
a different program for each n is required. This limitation can be
overcome if we restrict ourselves to the particular case in which
Xi = X for every i and some fixed X . Then

Q
i<n Xi amounts to

the type of finite lists of length n, and we write an algorithm that
works for lists of any length.

We first need to consider the particular case of the binary prod-
uct with types x0 = x and x1 = [x] for a given type x. Then
a pair (a0, a1) :: (x0, x1) can be coded as the list (a0 : a1) :: [x].
With these choices and coding, the above binary-product program
can be rewritten as

otimes :: J r x -> J r [x] -> J r [x]
otimes e0 e1 p = a0:a1
where a0 = e0(\x0 -> overline e1(\x1 -> p(x0:x1)))

a1 = e1(\x1 -> p(a0:x1))

This is now in a suitable form for iteration:

bigotimes :: [J r x] -> J r [x]
bigotimes [] = \p -> []
bigotimes (e:es) = e ‘otimes‘ bigotimes es

Expanding the definitions, this is the same algorithm given in Sec-
tion 1. Although we have motivated this algorithm by considering
finite lists, it does make sense for infinite lists too, as we shall see
in due course.

4. Playing games
We first show that products of selection functions compute optimal
plays and strategies. We then generalize the product of selection
functions in order to account for history dependent games, and
give concise and efficient implementations of Tic-Tac-Toe and n-
Queens as illustrations of the techniques.

4.1 Optimal outcomes, plays and strategies
As a first example, consider an alternating, two-person game that
finishes after exactly nmoves, with one of the players winning. The
i-th move is an element of the set Xi and the game is defined by a
predicate p :

Q
i<n Xi → R with

R = Bool,

that tells whether the first player, Eloise playing against Abelard,
wins a given play x = (x0, . . . , xn−1) ∈

Q
i<n Xi. Then Eloise

has a winning strategy for the game p if and only if

∃x0∈X0∀x1∈X1∃x2∈X2∀x3∈X3 · · · p(x0, . . . , xn−1).

If we define

φi =

(
∃Xi if i is even,
∀Xi if i is odd,

then this condition for Eloise having a winning strategy can be
equivalently expressed as O

i<n

φi

!
(p).

More generally, this value gives the optimal outcome of the game,
which takes place when all players play as best as they can. In
this example, the optimal outcome is True if Eloise has a winning
strategy, and False if Abelard has a winning strategy.

The following is proved in the paper [8], which also gives
formal definitions of the game theoretic terminology. Suppose each
quantifier φi has a selection function εi (thought of as a policy
function for the i-th move).



1. The sequence

a = (a0, . . . , an−1) =

 O
i<n

εi

!
(p)

is an optimal play.
This means that for every stage i < n of the game, the move ai

is optimal given that the moves a0, . . . , ai−1 have been played.

2. The function fk :
Q

i<k Xi → Xk defined by

fk(a) =

  
n−1O
i=k

εi

!
(λx.p(a++x))

!
0

is an optimal strategy for playing the game.
This means that if the sequence of moves a = (a0, . . . , ak−1)
have been played, then ak = fk(a) is at optimal move at
stage k.

Here k < n − 1, and x ∈
Qn−1

i=k Xi, and a++x is the
concatenation of the sequences a = (a0, . . . , ak−1) and x =
(xk, . . . , xn−1), and the subscript 0 picks the first element of
the list calculated inside the brackets.

As a second example, we choose

R = {−1, 0, 1}
instead, with the convention that −1 means that Abelard wins, 0
means that the game is a draw, and 1 that Eloise wins. Because
Eloise and Abelard want to maximize and minimize the outcome
of the game respectively, we replace the existential and universal
quantifiers sup and inf respectively,

φi =

(
supXi

if i is even,
infXi if i is odd.

The optimal outcome is still calculated as
N

i<n φi, which in this
case amounts to

sup
x0∈X0

inf
x1∈X1

sup
x2∈X2

inf
x3∈X3

· · · p(x0, . . . , xn−1),

and is 1 if Eloise has a winning strategy, −1 if Abelard has a
winning strategy, and 0 otherwise. Moreover, the above formulas
for computing optimal outcomes, plays and strategies apply.

4.2 History dependent games
In most sequential games of interest and that occur in practice, the
set of allowed moves at a given stage depends on the moves played
at the previous stages. The simplest case is that of a two-move
game. We assume that we have sets of moves X0 and X1. Once
a move x0 ∈ X0 has been played, the allowed moves form a set
Sx0 ⊆ X1 that depends on x0. We want to account for situations
such as

∀x0 ∈ X0∃x1 ∈ Sx0 p(x0, x1).

For example, in every pub there are a man a0 and a woman a1 older
than a0 such that if a0 buys a drink to a1, then every man buys
a drink to an older woman. Here Sx0 is the set of women older
than x0. This can be formalized by considering two quantifiers, the
second of which has a parameter:

φ0 ∈ KRX0, φ1 : X0 → KRX1.

In our running example,

φ0(q) = ∀x0 ∈ X0 q(x0),

φ1(x0)(q) = ∃x1 ∈ Sx0 q(x1).

Their history-dependent product is defined as the history-free prod-
uct considered before, taking care of instantiating the parameter

appropriately:

(φ0 ⊗ φ1)(p) = φ0(λx0.φ1(x0)(λx1.p(x0, x1))).

Similarly, given a selection function and a family of selection
functions,

ε0 ∈ JRX0, ε1 : X0 → JRX1,

we define their history-dependent product

(ε0 ⊗ ε1)(p) = (a0, a1)

where a0 = ε0(λx0.ε1(x0)(λx1.p(x0.x1)))

a1 = ε1(a0)(λx1.p(a0, x1)),

where it is understood that ε1(x0) = ε1(x0). And then again we
have

ε0 ⊗ ε1 = ε0 ⊗ ε1.
This amounts to saying that if ε0 is a selection function for the
quantifier φ0, and if ε1(x0) is a selection function for the quantifier
φ1(x0) for every x0 ∈ X0, then ε0 ⊗ ε1 is a selection function for
the quantifier φ0 ⊗ φ1.

We can iterate this if we are given a sequence of history depen-
dent selection functions

εn :
Y
i<n

Xi → JRXn.

We do this by induction, using the binary history dependent product
in the induction step: O

i<0

εi = λp.(),

O
i<n+1

εi = ε0()⊗ λx0.
O
i<n

(λ(x1, . . . , xi).εi+1(x0, . . . , xi)).

This can be written in Haskell as follows, if we trivialize the
dependent products as above:

otimes :: J r x -> (x -> J r [x]) -> J r [x]
otimes e0 e1 p = a0 : a1
where a0 = e0(\x0->overline(e1 x0)(\x1->p(x0:x1)))

a1 = e1 a0 (\x1 -> p(a0:x1))

bigotimes :: [[x] -> J r x] -> J r [x]
bigotimes [] = \p -> []
bigotimes (e:es) =
e[] ‘otimes‘ (\x->bigotimes[\xs->d(x:xs) | d<-es])

4.3 Implementation of games and optimal strategies
To define a sequential game, we need to define a type R of out-
comes, a type Move of moves, a predicate

p :: [Move] -> R

that gives the outcome of a play, and (history-dependent) selection
functions for each stage of the game:

epsilons :: [[Move] -> J R Move]

Once this is done, we can compute optimal plays, optimal out-
comes, and optimal strategies with the mathematical formulas
given above:

optimalPlay :: [Move]
optimalPlay = bigotimes epsilons p

optimalOutcome :: R
optimalOutcome = p optimalPlay

optimalStrategy :: [Move] -> Move



optimalStrategy as = head(bigotimes epsilons’ p’)
where epsilons’ = drop (length as) epsilons

p’ xs = p(as ++ xs)

Notice that all players use the same strategy function. To be pre-
cise, because the notion of player is not part of our definition of
sequential game, the strategy says what move should be played at
stage k given the moves at stages i < k. These three definitions ap-
ply to both history-free and history-dependent games and products
of selection functions.

4.4 Finite games of unbounded length
So far we have discussed finite products, and hence games, of fixed
length. In order to account for finite games of unbounded length,
we use infinite lazy lists, and we assume that the predicate p scans
the list of moves until the game ends and produces an outcome.
Of course, for this we need that every sequence of moves does
eventually lead to the end of the game in a finite number of moves.
The only explicitly given infinite lazy list in the specification of a
game is the list epsilons of policy functions for each stage of the
game.

4.5 Tic-Tac-Toe
We now consider Tic-Tac-Toe as an illustration of the above tech-
niques, but we emphasize that the ideas are very general and apply
to a wide variety of sequential games. The type of players is

data Player = X | O

The outcomes are 1 (first player wins), -1 (second player wins) and
0 (draw). We represent them by integers:

type R = Int

The possible moves are

0 1 2
3 4 5
6 7 8

So again we represent moves by integers:

type Move = Int

Although the concept of a game board doesn’t feature explicitly in
our specification of a sequential game, it is convenient to introduce
it in our implementation:

type Board = ([Move], [Move])

The components of the pair are the sets of moves played by X
and O respectively. We represent these sets as sorted lists without
repetitions for efficiency. A set of moves wins if it contains a row,
a column or a diagonal, and we use this to define the value of a
board:

wins :: [Move] -> Bool
wins =
someContained [[0,1,2],[3,4,5],[6,7,8],

[0,3,6],[1,4,7],[2,5,8],
[0,4,8],[2,4,6]]

value :: Board -> R
value (x,o) | wins x = 1

| wins o = -1
| otherwise = 0

Here the function

someContained :: Ord x => [[x]] -> [x] -> Bool

satisfies someContained xss ys = True if and only if some list
xs in the list xss has all elements occurring in the list ys. This is a

standard programming exercise, but we include the definition later
for the sake of completeness.

Next we define a function outcome such that, given a player, a
play (that is, a list of moves), and a board, produces the board that
results after playing the moves in an alternating fashion. If at some
point one player wins, then the remaining moves are ignored and
the board is unchanged:

outcome :: Player -> [Move] -> Board -> Board
outcome whoever [] board = board
outcome X (m : ms) (x, o) =
if wins o then (x, o)
else outcome O ms (insert m x, o)
outcome O (m : ms) (x, o) =
if wins x then (x, o)
else outcome X ms (x, insert m o)

Then the predicate that defines the game is obtained by computing
the outcome of the play starting from the empty board. We assume
that player X starts, as usual:

p :: [Move] -> R
p ms = value(outcome X ms ([],[]))

Finally, we need to the define the selection functions, or policy
functions, for each stage of the game. As discussed above, because
the player X wants to maximize the outcome of the game, and O
wants to minimize it, we use selection functions for the supremum
and infimum quantifiers, defined above. These selection functions
are history-dependent, where the history h is the sequence of moves
played so far, and they choose moves among those that haven’t been
played yet:

epsilons :: [[Move] -> J R Move]
epsilons = take 9 all
where all = epsilonX : epsilonO : all

epsilonX h = argsup ([0..8] ‘setMinus‘ h)
epsilonO h = arginf ([0..8] ‘setMinus‘ h)

The function

setMinus :: Ord x => [x] -> [x] -> [x]

returns a list with all elements that are in the first list but not in
the second one. And that’s it, apart from the standard definition
of set-theoretical operations, which are routine, but are given in
Section 4.7 below for the sake of completeness.

To test the program, we add

main :: IO ()
main =
putStr ("An optimal play for Tic-Tac-Toe is "
++ show optimalPlay ++ "\nand the optimal outcome is "
++ show optimalOutcome ++ "\n")

Compiling this with the Glasgow Haskell compiler as

$ ghc --make -O2 TicTacToe.hs

and running

$ time ./TicTacToe

under the operating system Ubuntu/Debian 9.10 in a 2.13GHz
machine, we get:

An optimal play for Tic-Tac-Toe is [0,4,1,2,6,3,5,7,8]
and the optimal outcome is 0

real 0m1.721s user 0m1.716s sys 0m0.004s

This means that, as is well known, if the two players play as best
as they can, the game is a draw. There are many optimal plays.



The computed one depends on which selection functions we have
chosen for our supremum and infimum quantifiers. In pictures, the
computed optimal play is:

X X
O

X X
O

X X O
O

X X O
O

X

X X O
O O
X

X X O
O O X
X

X X O
O O X
X O

X X O
O O X
X O X

4.6 n-Queens
We can solve the n-Queens puzzle using the same ideas. This time
this is a 1-player game, but again the number of players is not
explicit in the formalization of the problem and its solution. We
adopt the following conventions:

1. A solution is a permutation of [0..(n − 1)], which tells where
each queen should be placed in each row.

2. A move is an element of [0..(n−1)], saying in which column of
the given row (=stage of the game) the queen should be placed.

Let’s say we consider the standard chess board:

n = 8

The following should be self-explanatory given the above devel-
opment and assuming familiarity with the formulation of the n-
Queens problem:

type R = Bool
type Coordinate = Int
type Move = Coordinate
type Position = (Coordinate,Coordinate)

attacks :: Position -> Position -> Bool
attacks (x, y) (a, b) =

x == a || y == b || abs(x - a) == abs(y - b)

valid :: [Position] -> Bool
valid [] = True
valid (u : vs) =

not(any (\v -> attacks u v) vs) && valid vs

p :: [Move] -> R
p ms = valid(zip ms [0..(n-1)])

epsilons :: [[Move] -> J R Move]
epsilons = replicate n epsilon
where epsilon h = find ([0..(n-1)] ‘setMinus‘ h)

And that’s it. We test this as above:

main :: IO ()
main =

putStr ("An optimal play for " ++ show n
++ "-Queens is "
++ show optimalPlay
++ "\nand the optimal outcome is "
++ show optimalOutcome ++ "\n")

We then run

$ ghci --make -O2 NQueens.hs
$ time ./NQueens

and we get:

An optimal play for 8-Queens is [0,4,7,5,2,6,1,3]
and the optimal outcome is True

real 0m0.011s user 0m0.012s sys 0m0.000s

The fact that the optimal outcome is True means that the optimal
play is indeed a solution of the 8-Queens problem. With n = 12
the solution [0, 2, 4, 7, 9, 11, 5, 10, 1, 6, 8, 3] is computed in five
seconds. It should be apparent that many other standard games
and search problems can be concisely expressed and solved in this
formalism using the same pattern.

4.7 Appendix
For the sake of completeness, so that the reported experiments
can be reproduced by the readers, with similar run-time results
in similar machines, we include the routine code for finite sets
represented as sorted lists without repetitions used above:

contained :: Ord x => [x] -> [x] -> Bool
contained [] ys = True
contained xs [] = False
contained (us@(x : xs)) (y : ys)

| x == y = contained xs ys
| x >= y = contained us ys
| otherwise = False

someContained :: Ord x => [[x]] -> [x] -> Bool
someContained [] ys = False
someContained xss [] = False
someContained (xs : xss) ys
= contained xs ys || someContained xss ys

insert :: Ord x => x -> [x] -> [x]
insert x [] = [x]
insert x (vs@(y : ys))

| x == y = vs
| x < y = x : vs
| otherwise = y : insert x ys

delete :: Ord x => x -> [x] -> [x]
delete x [] = []
delete x (vs@(y : ys))

| x == y = ys
| x < y = vs
| otherwise = y : delete x ys

setMinus :: Ord x => [x] -> [x] -> [x]
setMinus xs [] = xs
setMinus xs (y : ys) = setMinus (delete y xs) ys

5. The Tychonoff Theorem
This section is about the close connection of some computational
and topological ideas, with applications to computation. For the
purposes of this exposition, it is not required to know what the
topological terms space, compact, continuous, countably based,
and Hausdorff mean. Rather, the readers should regard topology as
a foreign language, and use the facts stated below as a dictionary
between computational and topological notions. Computer-Science
Land readers interested in learning the language can start from
the introductory text [17], but this is not necessary in order to get
around in our guided tour to Topology Land.



5.1 An excursion to Topology Land
Call a set

1. searchable if it has a computable selection function, and

2. exhaustible if it has a computable boolean-valued quantifier.

It was shown in [7] that

Exhaustible sets are topologically compact.

Here we are implicitly assuming that we are dealing only with
sets of total elements. The situation for sets that include partial or
undefined elements is subtler and we refer the reader to [7]. By the
development of Section 2, using the monad morphism, we know
that searchable sets are exhaustible, and hence searchable sets are
compact too. The above quoted fact is related to the well-known
fact that

Computable functionals are topologically continuous.

This is so even from a purely operational point of view [4], without
considering any denotational semantics such as Scott domains. A
widely quoted topological slogan is that

Infinite compact sets behave, in many interesting and useful
ways, as if they were finite.

This matches computational intuition, because the ability to ex-
haustively search an infinite set, algorithmically and in finite time,
is indeed a computational sense in which the set behaves as if it
were finite. It may seem surprising at first sight that there are such
sets, but this was known in the 1950’s or before [13, 14], and we
shall see some examples shortly. We also need to mention that in
our computational setting:

Compact sets of total elements form countably based Haus-
dorff spaces.

Here we assume that any two equivalent total elements are iden-
tified (for example, the strict and non-strict constant function
Int -> Int are considered to be the same total element).

The papers [5, 7] explore what happens if one looks at theorems
in topology and applies this dictionary. Even theorems that are at
the core of topology and analysis turn out to be computationally
relevant:

1. Finite sets are compact, and hence for example the booleans are
compact.

2. Arbitrary products of compact sets are compact (Tychonoff
Theorem).
Hence the space of infinite sequences of booleans is compact.
This is known as the Cantor space, as it is topologically isomor-
phic (or homeomorphic) to the famous Cantor Third-Middle set
in the real line.

3. Continuous images of compact sets are compact.

4. Any non-empty, countably based, compact Hausdorff space is
a continuous image of the Cantor space.

Applying the dictionary, we get:

1. Finite sets are searchable, and hence for example the booleans
are searchable.

2. Finite and countably infinite products of searchable sets are
searchable.
Hence the Cantor space is searchable.

3. Computable images of searchable/exhaustible sets are search-
able/exhaustible.

4. Any non-empty exhaustible set is a computable image of the
Cantor space.

This is proved in [7], which develops more examples. Notice that
we have replaced compact sometimes by searchable and some-
times by exhaustible (and sometimes by both). We did this delib-
erately, in order to get the following souvenir from our topological
excursion:

Non-empty exhaustible sets are searchable.

This works as follows: if the set K is non-empty and exhaustible,
then it is a computable image of the Cantor space by (4), and
because the Cantor space is searchable by (1) and (2), the set K
is searchable by (3). This is computationally interesting, because
it says that if we have a search procedure that answers yes/no
(given by a quantifier), then we can automatically get a procedure
that gives witnesses (given by a selection function). Moreover, the
selection function can be obtained by a higher-type functional,
which can be written in Haskell with the following type:

selectionFromQuantifier :: K Bool x -> J Bool x

The definition and the argument that it works are rather complicated
and use non-trivial recursion-theoretic and topological technology,
and interested readers are referred to [7].

We now give some indications of how the above works, and de-
velop some examples. The finite and countably infinite Tychonoff
Theorem is implemented by the history-free version bigotimes of
the product of selection functions. Hence a selection function for
the Cantor space is given by

findCantor :: J Bool [Bool]
findCantor = bigotimes (repeat findBool)

We shall run an example in a moment. We now implement the fact
that computable images of searchable sets are searchable:

image :: (x -> y) -> J r x -> J r y
image f e = \q -> f(e(\x -> q(f x)))

This works as follows. Suppose that f :: x -> y is given and
that e :: J r x is a selection function for a set S ⊆ x, and let
f(S) = {f(s) | s ∈ S} ⊆ y be the image of S. Given a predicate
q :: y -> r, we wish to find s ∈ S such that q(s) = True.
The given algorithm first finds x such that q(f x) = True, using
the selection function e, and then applies f to it, giving the desired
result.

5.2 Deciding equality of functionals
As an application, perhaps contradicting common wisdom, we
write a total (!) functional that decides whether or not two given to-
tal functionals (Integer -> Bool) -> z are equivalent, where z
is any type with given decidable equality:

equal :: Eq z => ((Integer -> Bool) -> z)
-> ((Integer -> Bool) -> z) -> Bool

equal f g = foreveryFunction(\u->f u == g u)

This doesn’t contradict computability theory, which correctly as-
serts that, for example, equality of functions Integer -> Bool
is not decidable. A grammar that singles out infinitely many types
that have decidable equality is provided in [7].

To complete this definition, we first faithfully code functions
Integers->Bool as lazy lists by storing non-negative and nega-
tive arguments at even and odd indices respectively:

code :: (Integer -> Bool) -> [Bool]
code f = [f(reindex i)| i<-[0..]]

where reindex i | even i = i ‘div‘ 2
| otherwise = -((i+1)‘div‘ 2)



But actually we are interested in the opposite direction:

decode :: [Bool] -> (Integer -> Bool)
decode xs i | i >= 0 = xs ‘at‘ (i * 2)

| otherwise = xs ‘at‘ ((-i * 2) - 1)

at :: [x] -> Integer -> x
at (x:xs) 0 = x
at (x:xs) (n+1) = at xs n

Because the space of total infinite sequences [Bool] is searchable
by the computational Tychonoff Theorem, and because its image
under the computable function decode is the set of total functions
Integer -> Bool, this set of total functions is searchable and
hence exhaustible:

findFunction :: J Bool (Integer -> Bool)
findFunction = image decode findCantor
forsomeFunction :: K Bool (Integer -> Bool)
forsomeFunction = overline findFunction
foreveryFunction :: K Bool (Integer -> Bool)
foreveryFunction p = not(forsomeFunction(not.p))

This completes all ingredients needed to define the function equal.
Here is an experiment, where the function c is a coercion:

c :: Bool -> Integer
c False = 0
c True = 1

f, g, h :: (Integer -> Bool) -> Integer
f a = c(a(7 * c(a 4) + 4 * (c(a 7)) + 4))
g a = c(a(7 * c(a 5) + 4 * (c(a 7)) + 4))
h a = if not(a 7)

then if not(a 4) then c(a 4) else c(a 11)
else if a 4 then c(a 15) else c(a 8)

Are any two of these three functions equal? When we run this, using
the interpreter this time, we get:

$ ghci Tychonoff.hs
...
Ok, modules loaded: Main.
*Main> :set +s
*Main> equal f g
False
(0.02 secs, 4274912 bytes)
*Main> equal g h
False
(0.01 secs, 0 bytes)
*Main> equal f h
True
(0.00 secs, 0 bytes)

(We have no clue why the last computation is reported to take zero
bytes!) These examples are somewhat trivial, although probably
puzzling. A non-trivial application of quantification over the Cantor
space is given in [16]. This develops an algorithm for numerical in-
tegration, where real numbers are represented as infinite sequences
of digits. Hence there are no round-off errors in the produced re-
sults, which can be computed to arbitrary precision with guaran-
teed accuracy. Coming back to our trivial example, where do f and
g differ?

*Main> take 11 (code (findFunction(\u->g u /= h u)))
[True,True,True,True,True,True,
True,True,True,True,False]
(0.05 secs, 3887756 bytes)

We believe that these ideas open up the possibility of new, useful
tools for automatic program verification and bug finding.

5.3 Uniform continuity and the fan functional
Readers unfamiliar with topology and its relation to computation
are likely to be puzzled regarding the above decision procedure for
equality of functions, and more generally universal quantification
over infinite spaces. What is going on? Here is an indication.

Functions f :: [Bool] -> z are continuous. If z is topologi-
cally discrete (has equality), this means that for every input a, there
is an index n depending on a, such that the answer doesn’t depend
on positions of a with indices i ≥ n. This is intuitively clear from
a computational point of view, and can be rigorously proved (see
e.g. [4]). Now, the Cantor space is compact, and there are theorems
in topology that say that, in many cases, continuous functions de-
fined on compact spaces are uniformly continuous. In our case, this
means that there is a single index n, independent of the input, such
that the function f :: [Bool] -> z doesn’t look at indices i ≥ n
in order to produce the output. Another way of saying this is that if
two inputs a and b of the function f agree in the first n positions,
then they produce the same output. This criterion can be coded in
Haskell, and hence such an n can be computed. The functional that
computes it is called the fan functional, and is known since the
1950’s or even earlier [14]:

fan :: Eq z => ([Bool] -> z) -> Int
fan f = least(\n -> forevery(\a -> forevery(\b ->

agree n a b --> (f a == f b))))

least :: (Int -> Bool) -> Int
least p = if p 0 then 0 else 1+least(\n -> p(n+1))

forsome, forevery :: K Bool [Bool]
forsome = overline findCantor
forevery p = not(forsome(not.p))

agree :: Int -> [Bool] -> [Bool] -> Bool
agree n a b = take n a == take n b

(-->) :: Bool -> Bool -> Bool
p --> q = not p || q

Here is a trivial example:

*Main> fan(\a -> a !! 5 && a !! 6)
7

However, we are not interested in computing the fan functional.
What is relevant is that once we know the modulus of uniform
continuity n, it is enough to inspect 2n cases to figure out the
complete behaviour of the function.

If there is any magic in the above algorithm for quantifying over
the Cantor space and hence deciding equality of functions defined
on the Cantor space, it amounts to the facts that (1) this n is not
explicitly calculated by the quantification procedure (although it
can be calculated with the fan functional that uses this procedure),
and (2) very often, a small portion of the 2n cases actually need
to be checked in the quantification procedure (this has to do with
lazy evaluation). If we move from [Bool] to [Integer], then
compactness and uniform continuity fail, and moreover equality of
functions (Integer -> Integer) -> z is no longer decidable.

6. Monads
It turns out that the function image defined above is the functor of
a monad [8]. The unit is

singleton :: x -> J r x
singleton x = \p -> x

It implements the fact that singletons are searchable. The multipli-
cation is



bigunion :: J r (J r x) -> J r x
bigunion e = \p -> e(\d -> overline d p) p

It implements the fact that the union of a searchable set of search-
able sets is searchable (but it actually satisfies a more general spec-
ification involving generalized quantifiers [8]). To see that the mul-
tiplication behaves as claimed, assume that the selection function e
selects over a collection of selection functions d for sets Sd. Using
the selection function e, we find a selection function d such that
p(d p) holds, i.e. p s holds for some s ∈ Sd. Then we apply this
selection function to p, to actually find such s.

6.1 The continuation and selection monads
We now write the monads using Haskell’s conventions, where the
above gets a bit cumbersome and probably harder to understand,
as Haskell requires tags for monads, and hence we need functions
for extracting tags (namely quantifier and selection). We first
define the continuation monad:

module K (K(K), quantifier, unitK, functorK, muK) where

newtype K r x = K {quantifier :: (x -> r) -> r}

unitK :: x -> K r x
unitK x = K(\p -> p x)

functorK :: (x -> y) -> K r x -> K r y
functorK f phi = K(\q -> quantifier phi(\x -> q(f x)))

muK :: K r (K r x) -> K r x
muK phi = K(\p -> quantifier phi

(\gamma -> quantifier gamma p))

instance Monad (K r) where
return = unitK
phi >>= f = muK(functorK f phi)

Using this, we define the selection monad:

module J (J(J), selection, unitJ, functorJ, muJ,
module K, morphismJK) where

import K
newtype J r x = J {selection :: (x -> r) -> x}

morphismJK :: J r x -> K r x
morphismJK e = K(\p -> p(selection e p))

unitJ :: x -> J r x
unitJ x = J(\p -> x)

functorJ :: (x -> y) -> J r x -> J r y
functorJ f e = J(\q->f(selection e(\x->q(f x))))

muJ :: J r(J r x) -> J r x
muJ e = J(\p -> selection(selection e

(\d -> quantifier(morphismJK d) p)) p)

instance Monad (J r) where
return = unitJ
e >>= f = muJ(functorJ f e)

A proof that the monad laws hold is given in [8]. There is also
a short proof written in the dependently typed programming lan-
guage/proof checker Agda [3] available at [9].

6.2 Consequences of having a monad
First of all, now the history-free version of the functional bigotimes
is simply the Haskell standard prelude function sequence instan-
tiated to the selection monad:

sequence :: Monad m => [m a] -> m [a]
sequence = foldr mcons (return [])

where mcons p q =
p >>= \x->q >>= \y->return (x:y)

Thus, our computational manifestation of the Tychonoff Theorem
is already in the Haskell standard prelude, and becomes immedi-
ately available once one defines the selection monad.

Here the function mcons is simply one of the versions of our
function otimes generalized from the continuation and selection
monads to any monad, and can be equivalently written as

mcons :: Monad m => m x -> m[x] -> m[x]
xm ‘mcons‘ xsm =

do x <- xm
xs <- xsm
return (x:xs)

Moreover, sequence itself can be equivalently written as

sequence :: Monad m => [m x] -> m[x]
sequence [] = return []
sequence (xm : xms) =

do x <- xm
xs <- sequence xms
return(x : xs)

The history-dependent version of bigotimes also generalizes to
any monad:

hsequence :: Monad m => [[x] -> m x] -> m[x]
hsequence [] = return []
hsequence (xm : xms) =
do x <- xm []

xs <- hsequence[\ys -> ym(x:ys) | ym <- xms]
return(x : xs)

but this time this is not in the standard prelude.
Another consequence of having a monad is the topic of Sec-

tion 7, for which the remarks of Section 6.3 are relevant.

6.3 The function (h)sequence in monads other than J r

We observe that sequence and hsequence, although defined for
all monads, usually don’t produce convergent computations in
monads other than J r when supplied with infinite lists. More-
over, the type r has to have decidable equality for infinite products
of selection functions to be total (topologically, it has to be dis-
crete) [8]. The termination proof is non-trivial and relies on the
so-called bar induction principle. In particular, infinite products
of quantifiers cannot be computed with sequence or

N
, as also

shown in [8], with a continuity argument.
One can easily understand this kind of phenomenon with the list

monad. In the finite case, sequence computes cartesian products
in the lexicographic order:

Prelude> sequence [[0,1],[0,1],[0,1]]
[[0,0,0],[0,0,1],[0,1,0],[0,1,1],
[1,0,0],[1,0,1],[1,1,0],[1,1,1]]

But now the elements of the countable cartesian product {0, 1}ω
cannot be arranged in an infinite list, by Cantor’s diagonal argu-
ment, and when we attempt to list them using sequence, we get a
divergent computation, which in practice aborts for lack of mem-
ory:

Prelude> sequence (repeat [0,1])
*** Exception: stack overflow

It is reassuring to see Haskell refusing to give an answer to an
impossible question.



Apart from the selection monad, a monad for which we know
that sequence and hsequence converge for infinite lists is the
identity monad:

newtype Id x = Id { di :: x } deriving (Show)

instance Monad Id where
return a = Id a
(Id x) >>= f = f x

Here di removes the tag Id. Bearing in mind that semantically Id
is the identity, sequence essentially does nothing:

*Main> sequence [Id 1,Id 2,Id 3]
Id [1,2,3]

However, hsequence is much more interesting, as it amounts to
course-of-values recursion. The type of hsequence in this case
reduces to [[x] -> Id x] -> Id[x], and, according to the de-
velopment of the previous sections, the nth function in the in-
put list [f0, f1, f2, . . . ] is intended to have n arguments. What
hsequence computes, then, is the sequence Id[x0, x1, x2, . . . ] de-
fined by course-of-values induction as

Idxn = fn[x0, . . . , xn−1].

As an example, we enrich the literature with yet another way of
computing the Fibonacci sequence:

fibonacci :: [Integer]
fibonacci = di(hsequence (repeat f))

where f [ ] = Id 1
f [_] = Id 1
f xs = Id((xs !! (i - 1)) + (xs !! i))

where i = length xs - 1

Notice that the definition of f is not recursive. As discussed above,
the recursion is performed by hsequence. Here is what we get
when we run it:

*Main> take 10 fibonacci
[1,1,2,3,5,8,13,21,34,55]

Now, the identity functional and primitive recursion functional
are realizers of the intuitionistic axioms of choice and of depen-
dent choice respectively. It turns out that when the monad is J r
rather than Id, the functionals sequence and hsequence are in-
stead realizers of the classical axioms of choice and of (slightly
generalized) dependent choice, which brings us to the subject of
the next section.

7. The Double-Negation Shift
This last section of the tutorial is a rather brief excursion to Proof-
Theory Land with many gaps. We look at the J andK monads from
the propositions-as-types and proofs-as-programs point of view, for
which the dependently-typed functional language Agda [3] is more
appropriate than Haskell. We revert to the notation defined before
the previous section, in order to avoid the distracting tags that arise
in the definitions of the monads given in Section 6.1.

7.1 The Gödel–Gentzen negative translation and the
continuation monad

It is well known that KRA = ((A → R) → R) can be seen as
a generalized double-negation operator, reducing to standard dou-
ble negation when R = ⊥ (the proposition absurdity, or the empty
type). In order to avoid notational clutter, we write KA = KRA,
relying on the reader’s ability to infer the subscript R from the
context. The Gödel–Gentzen’s negative-translation prefixes dou-
ble negations in front of atomic propositions, disjunctions and ex-
istential quantifiers, leaving implications, conjunctions and univer-

sal quantifiers unchanged. A more general translation prefixes K
instead. Given a formula A, we denote its translation by AK . The
reason for considering this translation is that given any classical
proof ofA one can algorithmically find an intuitionistic proof of its
translation AK .

From the point of view of proofs-as-programs, this algorithm
amounts to one of the possible forms of the well-known continu-
ation passing style translation. If one works through the technical
details, one sees that what makes everything work is that fact thatK
is a monad. In particular, algebras of the monad, which are propo-
sitions A satisfying

KA→ A,

arise: every translated formula can be shown to be an algebra.
In the particular case R = ⊥, this amounts to saying that A
satisfies the double-negation elimination rule ¬¬A → A, which
is a classical principle. Thus, the translation forces this principle to
hold intuitionistically.

7.2 The Peirce translation, the selection monad and call/cc
One can routinely repeat the above development with any monad.
When this is done for JA = JRA = ((A→ R)→ A), algebras

JA→ A

are propositions that satisfy Peirce’s Law

((A→ R)→ A)→ A.

Usually the continuation monad is invoked to explain call/cc, but
a more natural explanation is obtained in terms of the selection
monad.

7.3 Extracting programs from proofs
Let A = ∀x∃y p(x, y) be a formula in Heyting arithmetic with
finite types (HAω), where p is decidable. Given an intuitionistic
proof of A one can find a program t (in Gödel system T , which
can be considered as a downgraded version of Haskell) such that
p(x, tx) holds for every x. One can think of this as follows: (1) p
is the specification of the input-output relation of a program to be
written down, (2) the given proof of A, because it is intuitionistic,
implicitly carries such a program, (3) there is a procedure that
exhibits the program given the proof, (4) hence rather than writing a
program, one can show in intuitionistic logic that for every input x
there is an output y satisfying the specification, and get a program
automatically, which by construction satisfies the specification.

Of course, this would be impractical for the average program-
mer. But there are many mathematicians and logicians in computer
science departments who are doing just that, using various kinds of
mechanical proof assistants. An added bonus is that there is a large
body of literature with intuitionistic proofs available, and hence po-
tentially a large body of programs that don’t need to be explicitly
written down. The downside is that formalizing rigorous proofs is
known to be no easy task. But nevertheless, if this activity is not
practical at the moment, it is certainly very exciting, scientifically
deep, enlightening, and mathematically pleasing.

But what if the proof of A is classical, rather than intuition-
istic, that is in Peano arithmetic with finite types (PAω)? Never
mind. Using the negative translation, one can still extract a pro-
gram. However, when one climbs up the logical systems, one gets
into difficulties. For example, there is no problem in using the ax-
ioms of choice or dependent choice in HAω , because they are real-
izable (one can write programs in system T that implement them).
But in PAω , the situation is subtler. What one needs is to realize the
negative translations of the axioms, which is problematic because
they involve existential quantification, which is altered by the trans-
lation.



7.4 The axiom of choice and the double-negation shift
Let T be any of the monads J orK. The T -translation of the axiom
of choice is

∀x T∃y A(x, y) =⇒ T∃f∀x A(x, f(x)).

The axiom of choice is the case in which T is the identity monad.
To extract programs from classical proofs in PAω using the axiom
of choice, one needs to realize the K-translation of the axiom of
choice, which is often referred to as the classical axiom of choice.
Spector’s idea [18] was to instead realize the T -shift (with T = K
and R = ⊥),

∀x TB(x) =⇒ T∀x B(x).

and prove that the intuitionistic axiom of choice together with this
gives the classical axiom of choice. This is enough to be able
to extract programs from proofs in PAω extended with choice.
Spector did this for the case where x ranges over natural numbers,
realizing the axiom of countable choice. Although Spector worked
with the dialectica interpretation and an extension of system T with
so-called bar recursion, his ideas remain relevant and crisp. See
also [1, 2].

It turns out that the history-free product of selection func-
tions directly realizes the T -shift for T = J , and that the J-
shift implies the K-shift when B is in the image of the K-
translation [11]. In fact, the iterated product of selection functions
provides a alternative and intuitive formulation of bar recursion.
Formal proofs/programs written in Agda are available at [9]. It is
worth mentioning that the J-shift is not system-T definable, and
that to define it in Agda one has to disable the termination checker
(and then users have to trust us).

7.5 The axiom of dependent choice
The previous subsection states that the history-free product of
selection functions realizes the J-shift, which in turn gives the
double-negation shift, and the J- and K-translations of the axiom
of choice. It turns out that the history-dependent product of selec-
tion functions realizes a version of the axiom of dependent choice.
We don’t have an implementation of this in Agda yet, but we are
working on this and related things.

7.6 Where are the (proofs and) programs for this section?
Proofs/programs, written in Agda [3], can be found at [9]. No pre-
vious knowledge of Agda is necessary: the types of the programs
read like usual logical expressions in ordinary mathematical nota-
tion, and their proofs look like usual functional programs. There
is also no need to run them, because their types tell us what their
behaviour will be. However, we plan to apply them to give an al-
ternative implementation of the program extractions from classical
proofs using dependent choice developed by Monika Seisenberger
in the system Minlog [15].

8. Concluding remarks
We have shown that diverse mathematical subjects coexist harmo-
niously and have a natural bed in functional programming: game
theory (optimal strategies), topology (Tychonoff Theorem), cat-
egory theory (monads), and proof theory (double-negation shift,
classical axiom of (dependent) choice).

An alternative title to this paper could have been selection func-
tions everywhere. It is the selection monad that unifies these math-
ematical subjects, where its associated product functional

N
com-

putes optimal strategies, implements a computational manifestation
of the Tychonoff Theorem, and realizes the double-negation shift
and the classical axiom of (dependent) choice.

Full mathematical proofs of the claims made in this tutorial can
be found in our joint papers given in the references. We wish to

point out that several of these proofs use mathematical machinery
outside the scope of this tutorial, but, as exemplified here, the
claims can be understood and applied with minimal mathematical
background if we accept them without proof.

Certainly more references are needed in a tutorial such as this:
please consult the references given in our self-references. We have
been unable to include them for lack of space.

